
1 
 Introduction
1.1 Purpose

This document describes a specification for a standard SNOMED CT Query Language that should be used to select concepts for inclusion in a simple concept reference set. The specification can be used to define value sets that may be bound to fields in information models, or to define sets of concepts to be included or excluded in particular use-cases.
1.2 Who Should Use this Specification?

The intended audience for this document includes technical professionals who are involved in the development or implementation of terminology systems or healthcare information systems that use SNOMED CT. This specification should be used in cases where it is required to create and maintain reference sets and value sets of SNOMED CT concepts that are based on intensional definitions. 

1.3 Scope of Specification

The scope of this document is to provide a language specification that should be used to define an expression that selects concepts for inclusion in a simple concept reference set.

The specification assumes that the expression is applied to a release of SNOMED CT, comprising an International version of SNOMED CT, optionally with one or more dependent extensions, and with one or more language reference sets selected. An expression can either be applied to an inferred view or to a stated view of a release.
The specification does not give guidance on how to describe which releases of SNOMED CT an expression can be used against, which extensions may be required, or which language reference sets may be selected. In addition, it does not provide guidance on how the resulting enumerated reference set (or the conditions under which it was generated) should be represented. In some cases, these points are implementation specific but, where relevant, additional guidance will be provided following approval of this specification.

1.4 Relationship to Compositional Grammar

The Query Specification Language and the Compositional Grammar meet different use cases. 

· The purpose of the Query Language is to enable the intensional selection of concepts for inclusion in a simple concept reference set.

· The purpose of the Compositional Grammar is to enable definition of a SNOMED CT post coordinated expression, by the combination of one or more pre-coordinated concepts.

1.5 Requirements

The following requirements have been used in producing this specification. The requirements were developed from experiences in developing and using query specification languages in the IHTSDO Workbench and other tools to develop small and large reference sets and from experiences in using and implementing RF2:

1. The expression language should allow the definition of a single expression that returns a set of concepts, and have no side effects.

2. The language should not be procedural, but comprise a set of nested functions.

3. It should be possible to nest the functions in any order, to aid ease of use. To achieve this, all functions should evaluate to a set of concepts, and have no side effects.

4. It should be possible to evaluate an expression against any release of SNOMED CT, with any number of extensions loaded, and any number of language reference sets selected. Although the expression language itself should not limit, define or make assumptions about the version of SNOMED CT to which it may be applied, these requirements can easily be supported by specific implementations, and guidance will be given on options to do this separately to this specification.

5. The expression language should be easy to use and easy to understand.

6. It should be possible to insert comments into queries following this specification, in order to aid understanding.

7. The expression language should aid efficiency; it should be possible for developers to create fast implementations of the language.

8. The language should support:

a. Set operations – union, intersection, exclusion of one set from another, etc.

b. Filter operations (for leaf nodes, primitive and fully defined concepts)

c. String matching operations

d. Selection of children and descendents, based on the SNOMED CT subsumption hierarchy

e. Selection of concepts having particular relationships (of a specific relationship type, to one or more concepts in a particular domain)

9. It should be possible to use previously defined or evaluated sets in the expression.

10. The language should be designed to accommodate future extensions. In particular, additional filtering options and advanced string matching operations may be desirable.

2 General Overview of the Specification
A reference set query expression can be held as a text string that may be edited using a text editor and saved in a text file, as a database string, or within an RF2 reference set. 

The following sections of this document provide a description of the syntax of the reference set specification language, and examples of its use. A reference set query expression can be evaluated to return the members of a single concept reference set. It has no side effects.

A reference set query expression can be evaluated against a particular release of SNOMED CT to produce a set of SNOMED CT concepts. This guide only describes the query specification language; a separate document will describe how a reference set specification can itself be stored within a reference set, along with associated metadata.
This query language is designed to work either on a snapshot of SNOMED CT, or on an ‘active only’ view of SNOMED CT. It has not been designed to work across histories, but can be used to produce delta and full views of reference sets, in conjunction with other processing. For more information about snapshot, delta and full views, please see the RF2 specification or the SNOMED CT Technical Implementation Guide.
3 Language Syntax

The table below lists all functions that form part of the reference set specification language. All the functions described below return a set of SNOMED CT concepts, and so the functions can be nested inside each other to produce complex specifications. Where functions are nested, the innermost functions should be evaluated first, before passing the resulting reference set members to the functions they are nested in. None of the functions have side effects.

The function definitions in the table below use the following conventions:

· A, B, C (i.e. – a capital letter) – a reference set or a SNOMED CT Identifier for a concept. For brevity, they are referred to as reference sets in the descriptions.

· s – a string literal, delimited by ‘|’ characters. 

· NamedRefset – A SNOMED CT Identifier for a concept denoting a reference set.

A SNOMED CT Identifier for a concept may be either 

· a numeric string of up to 18 characters (a SNOMED CT Identifier); or 

· a numeric string of up to 18 characters (a SNOMED CT Identifier) followed by a description (delimited by ‘|’ characters) that must match a description associated with the concept denoted by the SNOMED CT Identifier. 

Examples are given below:

· 307824009

· 307824009|Administrative statuses|

	Function
	Description 

	Intersection(A, B, C, ..)
	Include only concepts that are in reference set A, and that are also in reference set B and that are also in reference set C. A concept that is not in all of the input reference sets is not included in the output reference set. This function takes one or more arguments.

	FilterOnFullyDefined(A)
	Include only the concepts in reference set A that are Fully Defined concepts

	FilterOnPrimitive(A)
	Include only the concepts in reference set A that are Primitive concepts

	FilterOnLeaf(A)
	Include only the concepts in reference set A that are leaf concepts (i.e. - that don't have any children)

	FilterOnActive(A)
	Include only the concepts in reference set A that have a status of active.

	All
	Include all available Concepts (in SNOMED CT and any loaded extensions).

	FilterOnNoMatch(s, A)  
	Include only concepts in reference set A, that do not have an active description matching 's'. To be included in the resulting set, none of the concept’s active descriptions should match. See the next section for a definition of the matching criteria.

	FilterOnMatch(s, A)
	Include only concepts in reference set A, that have an active description matching ‘s'. To be included in the resulting set, at least one of the concept’s active descriptions should match. See the next section for a definition of the matching criteria.

	HasDirectRel(A, B)
	Include only concepts that have a relationship of a type within reference set A, to a target concept in reference set B. The relationship must be ungrouped, and must actually be against a concept, and not inferred, for that concept to be included.

	HasRel(A, B)
	Include only concepts that have a relationship of a type within reference set A, to a target concept in reference set B. The relationship must be ungrouped, and may be inferred from other relationships. 

	HasGroupedRels(A, B, C, D, …)
	Include only concepts that have a relationship group that includes a relationship of a type within reference set A, to a target concept in reference set B AND a relationship of a type within reference set C, to a target concept in reference set D. This function can take any even number of arguments. The grouped relationships may be inferred from other relationships.

	ChildrenAndSelf(A)        
	Include concepts in reference set A and also the direct children of concepts in reference set A

	Children(A)
	Include only the direct children of concepts in reference set A

	DescendantsAndSelf(A)
	Include concepts in reference set A and also their descendants 

	Descendants(A)
	Include only the descendants of concepts in reference set A

	Excludes(A, B)
	Include only concepts in reference set A that are not in reference set B

	MembersOf(NamedRefset)
	Include only the members of an already defined reference set. The reference set may either be a static reference set, that is an enumerated set of concepts, or a dynamic reference set that must be evaluated from a specification before being used.

	Union(A, B, C, ..)
	Include only members of A, B and C in the reference set. If a concept is in any of the input reference sets, it will be in the output reference set. Concepts are not duplicated in the output reference set. This function takes one or more arguments.


Notice that UpperCamelCase is used for function naming. In the descriptions above, it should also be noted that A, B and C can represent either a reference set or a single concept identifier. For brevity, they are referred to as reference sets in the descriptions.

Blank lines are ignored in expressions. Comments can be added to a reference set specification in one of two ways:

· using a double forward slash (‘//’), in which case the comment continues until the end of the line. 

· Using /* */ delimiters for multi-line comments, as follows:

/* this is 

   a multi-line

   comment */

4 Definition of Matching Criteria

This section defines the rules governing whether a match is made between ‘s’ and a concept in reference set A in the two functions FilterOnMatch(s, A) and FilterOnNoMatch(s, A).

· If ‘s’ matches any available active Description associated with a concept in reference set A, then that concept will be included in the results for FilterOnMatch(s, A), and not included in the results for FilterOnNoMatch(s, A).

· ‘s’ is a pipe delimited string of the form |word1 word2 … wordn|, where the string delimited by the pipe characters contains one or more words each separated by one or more white space characters. White space characters before the first word and after the last word are ignored.

· Each word can be of any length and take the form ‘abc’ or ‘abc*’.

· All words in ‘s’ must match words in the same description associated with a concept in reference set A in order for that concept to match.

· For word ‘abc’ to match on a Description, then ‘abc’ (in any case) must be a word in that Description (delimited by white space, or by the start/end of the description itself).

· For ‘abc*’ to match on a Description, then there must be a word in the Description that starts with ‘abc’ (in any case).

· Word matching is not case sensitive.

A few examples are given below. Given a concept in reference set A, as described in the first table, the results in the second table would be returned.

	Concept Id
	128302006

	Descriptions
	Chronic hepatitis C (disorder)

	
	Chronic hepatitis C

	
	Chronic type C viral hepatitis


	Function call
	Result

	FilterOnMatch(|chronic|, A)
	Included 

	FilterOnNoMatch(|chronic|, A)
	Excluded 

	FilterOnMatch(|Chr* |, A)
	Included

	FilterOnMatch(|CHR*|, A)
	Included

	FilterOnMatch(|chronic infection|, A)
	Excluded

	FilterOnMatch(|chr* hep*|, A)
	Included

	FilterOnMatch(|chr* c hepatitis|, A)
	Included

	FilterOnMatch(|hepatitis C (disorder)|, A)
	Included

	FilterOnMatch(|hepatitis type C (disorder)|, A)
	Excluded

	FilterOnNoMatch(|hepatitis type C (disorder)|, A)
	Included

	FilterOnMatch(|chr* vir* hep* C|, A)
	Included


Key
Included – concept 128302006 IS included in the resulting reference set.

Excluded - concept 128302006 is NOT included in the resulting reference set.

5 Examples of Query Expressions
The following examples give an indication of how the reference set query language specification can be used. 

// This query expression returns concepts in the Clinical finding sub-hierarchy

DescendantsAndSelf(404684003|Clinical finding|)

/* This query expression returns all fully defined concepts in the Clinical finding

 * sub-hierarchy */

FilterOnFullyDefined(DescendantsAndSelf(404684003|Clinical finding|))

/* This query expression returns the first three levels of the Clinical findings

 * hierarchy.

 */

ChildrenAndSelf(

 ChildrenAndSelf(
  ChildrenAndSelf(404684003|Clinical finding|)))

/* This query expression returns all concepts in the ‘Immune hypersensitivity

 * reaction hierarchy that have an explicit ungrouped ‘Causative agent’ relationship 

 * defined to any target concept.

 */

Intersection(

 DescendantsAndSelf(418925002|Immune hypersensitivity reaction|),

 HasDirectRel(246075003|Causitive agent|, All)

)

/* When run against an inferred view, this query expression returns all concepts 
 * that have a ‘Causative agent’ of ‘Prion’(as an ungrouped relationship). 
 * Note that concepts with inferred relationships are also returned. */

HasRel(246075003|Causitive agent|, 84676004|Prion|)

// This query returns all concepts in SNOMED CT and any loaded extensions.

All
/* This query expression returns all the members of the viral disease 

 * sub-hierarchy, together with members of a pre-defined reference set 

 * called “My Virus Refset”.

 */

Union(

 DescendantsAndSelf(34014006|Viral disease|),

 MembersOf(60140068|My Virus Reset|)

)

/* When run against the inferred view, this query expression returns all concepts 

 * that contain a group with a ‘Finding site’ of ‘Inguinal canal structure’ 

 * and an ‘Associated morphology’ of ‘Hermial opening’. 

 * Both these relationships must be in the same group. Note that concepts with inherited

 * grouped relationships are also returned.

 */

HasGroupedRels(363698007|Finding site|, 90785001|Inguinal canal structure|, 

               116676008|Associated morphology|, 414402003|Hermial opening|)

/* When run against the inferred view, this query expression returns all concepts 

 * that contain a first group with a ‘Finding site’ of ‘Inguinal canal structure’ 

 * and an ‘Associated morphology’ of ‘Hermial opening’, and a second group with

 * a ‘Finding site’ of ‘Abdominal cavity structure’ and an ‘Associated morphology’

 * of ‘Hernia’. Concepts with inherited grouped relationships are also returned.

 */

Intersection(

  HasGroupedRels(363698007|Finding site|, 90785001|Inguinal canal structure|, 

                 116676008|Associated morphology|, 414402003|Hermial opening|)

  HasGroupedRels(363698007|Finding site|, 52731004|Abdominal cavity structure|, 

                 116676008|Associated morphology|, 414403008|Hernia|))

/* This query expression returns all concepts in the ‘Viral disease’ sub-hierarchy

 * that have one or more descriptions containing all the words ‘chronic’, ‘hepatitis’,

 * and ‘c’.

 */

FilterOnMatch(|chronic c hepatitis|, DescendantsAndSelf(34014006|Viral disease|))

/* This query expression returns altered body structures, events, findings 

 * with explicit context, past history, family history and relevant clinical

 * findings. The expression returns about 105K concepts, when applied to the July

 * 2012 release of SNOMED CT.

 */

Union(

  Descendants(118956008|Body structure altered from its original|), 

  Excludes(

    Descendants(404684003|Clinical finding|),

    Union(

      DescendantsAndSelf(307824009|Administrative statuses|),

      DescendantsAndSelf(405533003|Adverse incident outcome categories|),

      DescendantsAndSelf(420134006|Propensity to adverse reactions|),

      DescendantsAndSelf(365858006|Prognosis/outlook finding|),

      DescendantsAndSelf(285153007|Sequelae of external causes and disorders|)

    )

  ),

  Descendants(272379006|Events|),

  Descendants(413350009|Finding with explicit context|),



  Descendants(57177007|Family history of|),

  Descendants(4908009|Past history of|)

) 

// this query expression returns concepts describing infectious arthritis

Intersection(Descendants(404684003|Clinical finding|),
 HasRel(116676008|Associated morphology|, DescendantsAndSelf(23583003|Inflammation|)),
 HasRel(363698007|Finding site|, DescendantsAndSelf(39352004|Joint structure|)),
 HasRel(246075003|Causative agent|, DescendantsAndSelf(410607006|Organism|)))

// this query expression returns allergens in a pre-defined specimen reference set
Intersection(MembersOf(4021000036102|Specimen type reference set|),

HasRel(370133003|Specimen substance|, Descendants(406455002|Allergen class|))) 

// this query expression returns diseases involving damage to acquired structures

Intersection(Descendants(64572001|Disease|),

 HasRel(116676008|Associated morphology|, DescendantsAndSelf(37782003|Damage|)),

 HasRel(363698007|Finding site|, 

  Excludes(DescendantsAndSelf(123037004|Body structure|), 

           DescendantsAndSelf(91723000|Anatomical structure|))))

// this query expression returns drugs containing amoxicillin and clavulanate
Intersection(Descendants(373873005|Pharmaceutical / biologic product|),

 HasRel(127489000|Has active ingredient|, DescendantsAndSelf(372687004|Amoxicillin|), 
 HasRel(127489000|Has active ingredient|, DescendantsAndSelf(395938000|Clavulanate|)))
// this query expression returns pediatric non-bacterial findings
Intersection(Descendants(404684003|Clinical finding|),

 HasRel(246454002|Occurrence|, Union(255398004|Childhood|, 3658006|Infancy|), 

 HasRel(246075003|Causative agent|, Exclude(All, Descendants(409822003|Bacteria|)))) 

6 Formal Definition of the Query Language

An Augmented Backus–Naur Form (ABNF) definition of the grammar is provided below:

expression = ws (intersection / filteronfullydefined / filteronprimitive / filteronleaf / filteronactive / all /  filteronexcludes / filteroncontains / hasrel / hasdirectrel / hasgroupedrels / childrenandself / children / descendantsandself / descendants / excludes / membersof / union / concept) ws

intersection = “Intersection” ws “(“ expression *(“,” expression) “)”

filteronfullydefined = “FilterOnFullyDefined” ws “(“ expression “)”

filteronprimitive = “FilterOnPrimitive” ws “(“ expression “)”

filteronleaf = “FilterOnLeaf” ws “(“ expression “)”

filteronactive = “FilterOnActive” ws “(“ expression “)”

all = “All”

filteronexcludes = “FilterOnExcludes” ws “(“ string “,” expression “)”

filteroncontains = “FilterOnContains” ws “(“string “,” expression “)”

hasrel = “HasRel” ws “(“ expression “,” expression “)”

hasdirectrel = “HasDirectRel” ws “(“ expression “,” expression “)”

hasgroupedrels = “HasGroupedRels” ws “(“ expression “,” expression *( “,” expression “,” expression ) “)”

childrenandself = “ChildrenAndSelf” ws “(“ expression “)”

children = “Children” ws “(“ expression “)”

descendantsandself = “DescendantsAndSelf” ws “(“ expression “)”

descendants = “Descendants” ws “(“ expression “)”

excludes = “Excludes” ws “(“ expression “,” expression “)”

membersof = “MembersOf” ws “(“ concept “)”

union = “Union” ws “(“ expression *(“,” expression) “)”

concept = ws conceptId [ string ] ws

conceptId = digitNonZero 5*17( digit ) ; between 5 and 17 digits are allowed
string = ws “|” ws stringVal ws “|” ws
stringVal = word *( 1*( SP / HTAB ) word )
digitNonZero = %x31-39 ;digits 1 through 9, but excluding 0

digit = %x30-39
word = 1*( %x21-7B / %x7D-7E / UTF8-2 / UTF8-3 / UTF8-4 ) ; exclude pipe character and ws

ws = *( SP / HTAB / CR / LF / comment) ; white space or comment

comment = ( “//” commentchars CR LF ) / ( “/*” anychars “*/” ) ; single line or multi-line comments
commentchars = *( HTAB / %x20-7E / UTF8-2 / UTF8-3 / UTF8-4 )  ;not CR or LF

anychars = *( HTAB / %x20-7E / UTF8-2 / UTF8-3 / UTF8-4 / CR / LF )

SP = %x20

HTAB = %x09

CR = %x0D

LF = %x0A

UTF8-2 = %xC2-DF UTF8-tail

UTF8-3 = %xE0 %xA0-BF UTF8-tail / %xE1-EC 2( UTF8-tail ) / %xED %x80-9F UTF8-tail / 
%xEE-EF 2( UTF8-tail )

UTF8-4 = %xF0 %x90-BF 2( UTF8-tail ) / %xF1-F3 3( UTF8-tail ) / %xF4 %x80-8F 2( UTF8-tail )

UTF8-tail = %x80-BF

This definition may be used to assess conformance of a query expression against the query language specification.

